
Gramáticas libres de contexto: Ejemplo

x Ejemplos de frases

u El gato come pescado

u El perro come carne

x Ejemplo de gramática libre de contexto (GLC)

oración --> sintagma_nominal, sintagma_verbal
sintagma_nominal --> nombre
sintagma_nominal --> artı́culo, nombre
sintagma_verbal --> verbo, sintagma_nominal
artı́culo --> [el]
nombre --> [gato]
nombre --> [perro]
nombre --> [pescado]
nombre --> [carne]
verbo --> [come]

Gramáticas libres de contexto: Árbol de análisis

x Árbol de análisis

pescado

nombre

sintagma_nomimal

comegato

verbo

sintagma_verbal

nombrearticulo

el

sintagma_nominal

oracion

Gramáticas libres de contexto: Definiciones

x Concepto de gramática: G = (N,T,P,S)

u N: vocabulario no terminal (categoŕıas sintácticas)

u T: vocabulario terminal

u P: reglas de producción

u S: śımbolo inicial

x Vocabulario: V = N ∪ T es el vocabulario con N ∩ T = ∅
x Derivaciones

u xAy =⇒ xwy mediante A =⇒ w

u x
∗

=⇒ y si existen x1, x2, . . . , xn tales que x = x1 =⇒ x2 · · · =⇒ xn−1 =⇒ xn = y

x Lenguaje definido por una gramática: L(G) = {x ∈ T ∗ : S
∗

=⇒ x}
x Gramáticas libres de contextos (GLC): A =⇒ w con A ∈ N y w ∈ V ∗

Reconocedor de GLC mediante append

x Representación de oraciones en Prolog

[el, gato, come, pescado] [el, perro, come, carne]

x Reconocedor de GLC en Prolog mediante append

u Sesión (con coste)

?- time(oración([el,gato,come,pescado])).
% 178 inferences in 0.00 seconds (Infinite Lips)
Yes
?- time(oración([el,come,pescado])).
% 349 inferences in 0.00 seconds (Infinite Lips)
No

u Definición

oración(O) :- sintagma_nominal(SN), sintagma_verbal(SV), append(SN,SV,O).
sintagma_nominal(SN) :- nombre(SN).
sintagma_nominal(SN) :- artı́culo(A), nombre(N), append(A,N,SN).
sintagma_verbal(SV) :- verbo(V), sintagma_nominal(SN), append(V,SN,SV).

artı́culo([el]).
nombre([gato]). nombre([perro]). nombre([pescado]). nombre([carne]).
verbo([come]).

Reconocedor de GLC mediante append

x Otros usos de la gramática

u Generación de las oraciones

?- oración(O).
O = [gato, come, gato] ;
O = [gato, come, perro] ;
O = [gato, come, pescado]
Yes
?- findall(_O,oración(_O),_L),length(_L,N).
N = 64

u Reconocedor de las categoŕıas gramaticales

?- sintagma_nominal([el,gato]).
Yes
?- sintagma_nominal([un,gato]).
No

u Generador de las categorias gramaticales

?- findall(_SN,sintagma_nominal(_SN),L).
L = [[gato],[perro],[pescado],[carne],[el,gato],[el,perro],[el,pescado],[el,carne]]

Reconocedor de GLC mediante listas de diferencia

x Reconocedor de GLC en Prolog mediante listas de diferencia

u Sesión (y ganancia en eficiencia)

?- time(oración([el,gato,come,pescado]-[])).
% 9 inferences in 0.00 seconds (Infinite Lips)
Yes
?- time(oración([el,come,pescado]-[])).
% 5 inferences in 0.00 seconds (Infinite Lips)
No

u Definición
oración(A-B) :- sintagma_nominal(A-C), sintagma_verbal(C-B).
sintagma_nominal(A-B) :- nombre(A-B).
sintagma_nominal(A-B) :- artı́culo(A-C), nombre(C-B).
sintagma_verbal(A-B) :- verbo(A-C), sintagma_nominal(C-B).

artı́culo([el|A]-A).
nombre([gato|A]-A).
nombre([perro|A]-A).
nombre([pescado|A]-A).
nombre([carne|A]-A).
verbo([come|A]-A).

PD 2003–04 CcIa Procesamiento de lenguaje natural 9.7



Reconocedor de GLC mediante listas de diferencia

x Otros usos de la gramática

u Generación de las oraciones

?- oración(O-[]).
O = [gato, come, gato] ;
O = [gato, come, perro] ;
O = [gato, come, pescado]
Yes
?- findall(_O,oración(_O-[]),_L),length(_L,N).
N = 64

u Reconocedor de las categoŕıas gramaticales

?- sintagma_nominal([el,gato]-[]).
Yes
?- sintagma_nominal([un,gato]-[]).
No

u Generador de las categorias gramaticales

?- findall(_SN,sintagma_nominal(_SN-[]),L).
L = [[gato],[perro],[pescado],[carne],[el,gato],[el,perro],[el,pescado],[el,carne]]

Gramáticas de cláusulas definidas: Ejemplo

x Ejemplo de GCD

u Definición

oración --> sintagma_nominal, sintagma_verbal.
sintagma_nominal --> nombre.
sintagma_nominal --> artı́culo, nombre.
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo --> [el].
nombre --> [gato].
nombre --> [perro].
nombre --> [pescado].
nombre --> [carne].
verbo --> [come].

Gramáticas de cláusulas definidas: Usos

x Usos de la gramática

u Reconocimiento de oraciones

?- oración([el,gato,come,pescado],[]).
Yes
?- oración([el,come,pescado],[]).
No

u Generación de las oraciones

?- oración(O,[]).
O = [gato, come, gato] ;
O = [gato, come, perro] ;
O = [gato, come, pescado]
Yes
?- findall(_O,oración(_O,[]),_L),length(_L,N).
N = 64

Gramáticas de cláusulas definidas: Usos

u Reconocedor de las categoŕıas gramaticales

?- sintagma_nominal([el,gato],[]).
Yes
?- sintagma_nominal([un,gato],[]).
No

u Generador de las categorias gramaticales

?- findall(_SN,sintagma_nominal(_SN,[]),L).
L = [[gato],[perro],[pescado],[carne],[el,gato],[el,perro],[el,pescado],[el,carne]]

u Determinacion de elementos
?- oración([X,gato,Y,pescado],[]).
X = el
Y = come ;
No

u La relación phrase

?- phrase(oración,[el,gato,come,pescado]).
Yes
?- phrase(sintagma_nominal,L).
L = [gato] ;
L = [perro]
Yes

Gramáticas de cláusulas definidas: Compilación

u Compilación

?- listing([oración,sintagma_nominal,sintagma_verbal,artı́culo,nombre,verbo]).
oración(A, B) :- sintagma_nominal(A, C), sintagma_verbal(C, B).
sintagma_nominal(A, B) :- nombre(A, B).
sintagma_nominal(A, B) :- artı́culo(A, C), nombre(C, B).
sintagma_verbal(A, B) :- verbo(A, C), sintagma_nominal(C, B).
artı́culo([el|A], A).
nombre([gato|A], A).
nombre([perro|A], A).
nombre([pescado|A], A).
nombre([carne|A], A).
verbo([come|A], A).
Yes

u Eficiencia

?- time(oración([el,gato,come,pescado],[])).
% 9 inferences in 0.00 seconds (Infinite Lips)
Yes
?- time(oración([el,come,pescado],[])).
% 5 inferences in 0.00 seconds (Infinite Lips)
No

Reglas recursivas en GCD

x Problema: Extender el ejemplo de GCD para aceptar oraciones como
[el,gato,come,pescado,o,el,perro,come,pescado]

x Primera propuesta

u GCD

oración --> oración, conjunción, oración.
oración --> sintagma_nominal, sintagma_verbal.
sintagma_nominal --> nombre.
sintagma_nominal --> artı́culo, nombre.
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo --> [el].
nombre --> [gato].
nombre --> [perro].
nombre --> [pescado].
nombre --> [carne].
verbo --> [come].
conjunción --> [y].
conjunción --> [o].

Reglas recursivas en GCD

u Sesión

?- oración([el,gato,come,pescado,o,el,perro,come,pescado],[]).
ERROR: Out of local stack

?- listing(oración).
oración(A, B) :-

oración(A, C),
conjunción(C, D),
oración(D, B).

oración(A, B) :-
sintagma_nominal(A, C),
sintagma_verbal(C, B).

Yes

Reglas recursivas en GCD

x Segunda propuesta

oración --> sintagma_nominal, sintagma_verbal.
oración --> oración, conjunción, oración.
sintagma_nominal --> nombre.
sintagma_nominal --> artı́culo, nombre.
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo --> [el].
nombre --> [gato].
nombre --> [perro].
nombre --> [pescado].
nombre --> [carne].
verbo --> [come].
conjunción --> [y].
conjunción --> [o].

u Sesión

?- oración([el,gato,come,pescado,o,el,perro,come,pescado],[]).
Yes
?- oración([un,gato,come],[]).
ERROR: Out of local stack

Reglas recursivas en GCD

x Tercera propuesta

u GCD
oración --> oración_simple.
oración --> oración_simple, conjunción, oración.
oración_simple --> sintagma_nominal, sintagma_verbal.
sintagma_nominal --> nombre.
sintagma_nominal --> artı́culo, nombre.
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo --> [el].
nombre --> [gato].
nombre --> [perro].
nombre --> [pescado].
nombre --> [carne].
verbo --> [come].
conjunción --> [y].
conjunción --> [o].

u Sesión

?- oración([el,gato,come,pescado,o,el,perro,come,pescado],[]).
Yes
?- oración([un,gato,come],[]).
No

GCD para un lenguaje formal

x GCD para el lenguaje formal {anbn : n ∈ N}
u Sesión

?- s([a,a,b,b],[]).
Yes
?- s([a,a,b,b,b],[]).
No
?- s(X,[]).
X = [] ;
X = [a, b] ;
X = [a, a, b, b] ;
X = [a, a, a, b, b, b]
Yes

u GCD

s --> [].
s --> i,s,d.
i --> [a].
d --> [b].

Árbol de análisis con GCD

x Árbol de análisis con GCD
u Sesión

?- oración(A,[el,gato,come,pescado],[]).
A = o(sn(art(el),n(gato)),sv(v(come),sn(n(pescado))))
Yes

u Definición

oración(o(SN,SV)) --> sintagma_nominal(SN), sintagma_verbal(SV).
sintagma_nominal(sn(N)) --> nombre(N).
sintagma_nominal(sn(Art,N)) --> artı́culo(Art), nombre(N).
sintagma_verbal(sv(V,SN)) --> verbo(V), sintagma_nominal(SN).
artı́culo(art(el)) --> [el].
nombre(n(gato)) --> [gato].
nombre(n(perro)) --> [perro].
nombre(n(pescado)) --> [pescado].
nombre(n(carne)) --> [carne].
verbo(v(come)) --> [come].

Árbol de análisis con GCD

u Compilación

?- listing([oración,sintagma_nominal,sintagma_verbal,artı́culo,nombre,verbo]).

oración(o(A, B), C, D) :- sintagma_nominal(A, C, E), sintagma_verbal(B, E, D).

sintagma_nominal(sn(A), B, C) :- nombre(A, B, C).
sintagma_nominal(sn(A, B), C, D) :- artı́culo(A, C, E), nombre(B, E, D).

sintagma_verbal(sv(A, B), C, D) :- verbo(A, C, E), sintagma_nominal(B, E, D).

artı́culo(art(el), [el|A], A).

nombre(n(gato), [gato|A], A).
nombre(n(perro), [perro|A], A).
nombre(n(pescado), [pescado|A], A).
nombre(n(carne), [carne|A], A).

verbo(v(come), [come|A], A).

Yes

Concordancia de género en GCD

x Concordancia de género en GCD

u Sesión

?- oración([el,gato,come,pescado],[]).
Yes
?- oración([la,gato,come,pescado],[]).
No
?- oración([la,gata,come,pescado],[]).
Yes

u Definición

oración --> sintagma_nominal, sintagma_verbal.
sintagma_nominal --> nombre(_).
sintagma_nominal --> artı́culo(G), nombre(G).
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo(masculino) --> [el].
artı́culo(femenino) --> [la].
nombre(masculino) --> [gato].
nombre(femenino) --> [gata].
nombre(masculino) --> [pescado].
verbo --> [come].

Concordancia de número en GCD

x Concordancia de número en GCD
u Sesión

?- oración([el,gato,come,pescado],[]).
Yes
?- oración([los,gato,come,pescado],[]).
No
?- oración([los,gatos,comen,pescado],[]).
Yes

u Definición

oración --> sintagma_nominal(N), sintagma_verbal(N).
sintagma_nominal(N) --> nombre(N).
sintagma_nominal(N) --> artı́culo(N), nombre(N).
sintagma_verbal(N) --> verbo(N), sintagma_nominal(_).
artı́culo(singular) --> [el]. artı́culo(plural) --> [los].
nombre(singular) --> [gato]. nombre(plural) --> [gatos].
nombre(singular) --> [perro]. nombre(plural) --> [perros].
nombre(singular) --> [pescado].
nombre(singular) --> [carne].
verbo(singular) --> [come]. verbo(plural) --> [comen].

Ejemplo de GCD no GCL

x GCD para el lenguaje formal {anbncn : n ∈ N}
u Sesión

?- s([a,a,b,b,c,c],[]).
Yes
?- s([a,a,b,b,b,c,c],[]).
No
?- s(X,[]).
X = [] ;
X = [a, b, c] ;
X = [a, a, b, b, c, c]
Yes

u GCD

s --> bloque_a(N), bloque_b(N), bloque_c(N).
bloque_a(0) --> [].
bloque_a(suc(N)) --> [a], bloque_a(N).
bloque_b(0) --> [].
bloque_b(suc(N)) --> [b], bloque_b(N).
bloque_c(0) --> [].
bloque_c(suc(N)) --> [c], bloque_c(N).

GCD con llamadas a Prolog

x GCD para el lenguaje formal L = {a2nb2nc2n : n ∈ N}
u Ejemplos

?- s([a,a,b,b,c,c],[]).
Yes
?- s([a,b,c],[]).
No
?- s(X,[]).
X = [] ; X = [a,a,b,b,c,c] ; X = [a,a,a,a,b,b,b,b,c,c,c,c] ; X = [a,a,a,a,a,a,b,b,b,b,b,b,c,c,c,c,c,c]
Yes

u GCD
s --> bloque_a(N), bloque_b(N), bloque_c(N), {par(N)}.
bloque_a(0) --> [].
bloque_a(s(N)) --> [a],bloque_a(N).
bloque_b(0) --> [].
bloque_b(s(N)) --> [b],bloque_b(N).
bloque_c(0) --> [].
bloque_c(s(N)) --> [c],bloque_c(N).

par(0).
par(s(s(N))) :- par(N).

Separación de reglas y lexicón

x Gramática
u Lexicón
lex(el,artı́culo).
lex(gato,nombre).
lex(perro,nombre).
lex(pescado,nombre).
lex(carne,nombre).
lex(come,verbo).

u Regla
oración --> sintagma_nominal, sintagma_verbal.
sintagma_nominal --> nombre.
sintagma_nominal --> artı́culo, nombre.
sintagma_verbal --> verbo, sintagma_nominal.
artı́culo --> [Palabra], {lex(Palabra,artı́culo)}.
nombre --> [Palabra], {lex(Palabra,nombre)}.
verbo --> [Palabra], {lex(Palabra,verbo)}.

u Sesión
?- oración([el,gato,come,pescado],[]).
Yes
?- oración([el,come,pescado],[]).
No

Separación de reglas y lexicón con concordancia

u Sesión

?- oración([el,gato,come,pescado],[]). ==> Yes
?- oración([los,gato,come,pescado],[]). ==> No
?- oración([los,gatos,comen,pescado],[]). ==> Yes

u Lexicón

lex(el,artı́culo,singular). lex(los,artı́culo,plural).
lex(gato,nombre,singular). lex(gatos,nombre,plural).
lex(perro,nombre,singular). lex(perros,nombre,plural).
lex(pescado,nombre,singular). lex(pescados,nombre,plural).
lex(carne,nombre,singular). lex(carnes,nombre,plural).
lex(come,verbo,singular). lex(comen,verbo,plural).

u Reglas

oración --> sintagma_nominal(N), sintagma_verbal(N).
sintagma_nominal(N) --> nombre(N).
sintagma_nominal(N) --> artı́culo(N), nombre(N).
sintagma_verbal(N) --> verbo(N), sintagma_nominal(_).
artı́culo(N) --> [Palabra], {lex(Palabra,artı́culo,N)}.
nombre(N) --> [Palabra], {lex(Palabra,nombre,N)}.
verbo(N) --> [Palabra], {lex(Palabra,verbo,N)}.

Lexicón con género y número

x Lexicón con género y número

u Sesión

?- oración([la,profesora,lee,un,libro],[]). ==> Yes
?- oración([la,profesor,lee,un,libro],[]). ==> No
?- oración([los,profesores,leen,un,libro],[]). ==> Yes
?- oración([los,profesores,leen],[]). ==> Yes
?- oración([los,profesores,leen,libros],[]). ==> Yes

u Lexicón

lex(el,determinante,masculino,singular). lex(los,determinante,masculino,plural).
lex(la,determinante,femenino,singular). lex(las,determinante,femenino,plural).
lex(un,determinante,masculino,singular). lex(una,determinante,femenino,singular).
lex(unos,determinante,masculino,plural). lex(unas,determinante,femenino,plural).

lex(profesor,nombre,masculino,singular). lex(profesores,nombre,masculino,plural).
lex(profesora,nombre,femenino,singular). lex(profesoras,nombre,femenino,plural).
lex(libro,nombre,masculino,singular). lex(libros,nombre,masculino,plural).

lex(lee,verbo,singular). lex(leen,verbo,plural).

Lexicón con género y número

u Reglas

oración --> sintagma_nominal(N), verbo(N), complemento.
complemento --> [].
complemento --> sintagma_nominal(_).
sintagma_nominal(N) --> nombre(_,N).
sintagma_nominal(N) --> determinante(G,N), nombre(G,N).
determinante(G,N) --> [P],{lex(P,determinante,G,N)}.
nombre(G,N) --> [P],{lex(P,nombre,G,N)}.
verbo(N) --> [P],{lex(P,verbo,N)}.

